SPSS Syntax for Matrix Algebra
*  The data in the variables is

var00001, var00002,var00003, var00004

matrix.  /*example of computing determinant

get x /variables var00001, var00002. /* this creates a matrix with the n rows and  p=2 columns

get y /variables var00003, var00004. /* this creates a matrix with the n rows and  p=2 columns

print x.  /*printing lets us see the outcome of an action

print y.

compute detx=det(x).  /* compute determinant of x

compute dety=det(y).

print detx.

print dety.

end matrix.

Run MATRIX procedure:

X

  2  1

  7  3

Y

  1  1

  3  7

DETX

  -1.000000000

DETY

  4

------ END MATRIX -----

matrix.  /* example of matrix multiplication that verifies that xy is not always yx

get x /variables var00001, var00002. /* this creates a matrix with the n rows and  p=2 columns

get y /variables var00003, var00004. /* this creates a matrix with the n rows and  p=2 columns

compute xtimesy=x*y. /*multiple x time y in that order

compute ytimesx=y*x.

print xtimesy.

print ytimesx.

end matrix.

Run MATRIX procedure:

XTIMESY

   5   9

  16  28

YTIMESX

   9   4

  55  24

------ END MATRIX -----

matrix. /* computing eigenvalues and eigenvectors

get x /variables var00001, var00002. /* this creates a matrix with the n rows and  p=2 columns

get y /variables var00003, var00004. /* this creates a matrix with the n rows and  p=2 columns

compute xtx=transpos(x)*x. /* compute x'x which is a symmetric matrix; note: "transpos" could be shortened to just "t"

print xtx.

call eigen(xtx,eigvec,eigval). /*compute eigenvalues and eigenvectors of x'x

print eigval.

print eigvec.

/* the original matrix x'x can be represented approximately using the "spectral decomposition" of eigenvalues and eigenvectors

compute approx1=eigval(1)*eigvec(:,1)*t(eigvec(:,1)).

print approx1.

compute approx2=eigval(1)*eigvec(:,1)*t(eigvec(:,1))+eigval(2)*eigvec(:,2)*t(eigvec(:,2)).

print approx2.   /* the approximation with only the largest eigenvalue is not bad, but with both it is perfect 

end matrix.

Run MATRIX procedure:

XTX

  53  23

  23  10

EIGVAL

   62.98412298

     .01587702

EIGVEC

   .9173014439  -.3981934969

   .3981934969   .9173014439

APPROX1

_

   52.99748257   23.00579929

   23.00579929    9.98664041

APPROX2

   53.00000000   23.00000000

   23.00000000   10.00000000

------ END MATRIX -----

*read in Online Shopping Attitude mini dataset

q01_bk
q01_sh
q01_tp

/* Eigenvalues and eigenvectors of correlation matrix R

matrix.  

get x /variables q01_bk,q01_sh,q01_tp,q01_dd,q01_fr,q01_fd. /* this creates a matrix with the n rows and  p=6 columns

compute n=nrow(x).

compute one=make(n,1,1).

compute i=ident(n).

compute xbar=t(x)*one/n.

Compute h=i-one*t(one)/n.  /* centering matrix

compute e=h*x. /* observations as deviations from mean

compute s=T(x)*h*x/(n-1) . /* covariance matrix

print s/format="F5.2".

compute d=mdiag(sqrt(diag(s))).

compute R=inv(d)*s*inv(d). /*correlation matrix

print R/format="F5.2".

call eigen(R,P,L).

print L/format="F5.2".

print P/format="F5.2".

compute G=P*mdiag(L)*T(P)./*complete spectral decomposition

print G/format="F5.2"/title="PlambdaP'".

compute q=P(:,1:4). /*Use only first 4 of 6 eigenvectors in spectral approximation of R

print q/format="F5.2".

compute m=L(1:4).

print m/format="F5.2".

compute g4=q*mdiag(m)*t(q).

print R/format="F5.2".

print g4/title="Approximation of R with 4 of 6 eigenvalues"/format="F5.2".

end matrix.

Run MATRIX procedure:

S

  3.16   .91  2.13  2.27  2.47  1.04

   .91  4.46   .46  1.52  2.51  1.69

  2.13   .46  3.34  2.30  2.40  1.80

  2.27  1.52  2.30  4.10  4.36  1.38

  2.47  2.51  2.40  4.36  5.51  1.36

  1.04  1.69  1.80  1.38  1.36  1.82

R

  1.00   .24   .66   .63   .59   .44

   .24  1.00   .12   .36   .51   .59

   .66   .12  1.00   .62   .56   .73

   .63   .36   .62  1.00   .92   .50

   .59   .51   .56   .92  1.00   .43

   .44   .59   .73   .50   .43  1.00

L

  3.68

  1.00

   .78

   .41

   .12

   .01

P

  -.40  -.34   .02  -.84   .12  -.09

  -.29   .81   .08  -.26  -.21   .38

  -.42  -.36  -.46   .22  -.52   .41

  -.46  -.14   .39   .33   .59   .41

  -.45   .01   .51   .23  -.43  -.55

  -.40   .28  -.61   .17   .38  -.46

PlambdaP'

  1.00   .24   .66   .63   .59   .44

   .24  1.00   .12   .36   .51   .59

   .66   .12  1.00   .62   .56   .73

   .63   .36   .62  1.00   .92   .50

   .59   .51   .56   .92  1.00   .43

   .44   .59   .73   .50   .43  1.00

Q

  -.40  -.34   .02  -.84

  -.29   .81   .08  -.26

  -.42  -.36  -.46   .22

  -.46  -.14   .39   .33

  -.45   .01   .51   .23

  -.40   .28  -.61   .17

M

  3.68

  1.00

   .78

   .41

R

  1.00   .24   .66   .63   .59   .44

   .24  1.00   .12   .36   .51   .59

   .66   .12  1.00   .62   .56   .73

   .63   .36   .62  1.00   .92   .50

   .59   .51   .56   .92  1.00   .43

   .44   .59   .73   .50   .43  1.00

Approximation of R with 4 of 6 eigenvalues

  1.00   .25   .66   .62   .60   .43

   .25   .99   .10   .37   .50   .60

   .66   .10   .97   .65   .54   .75

   .62   .37   .65   .96   .95   .48

   .60   .50   .54   .95   .98   .44

   .43   .60   .75   .48   .44   .98

------ END MATRIX -----

/*Singular decomposition of A into ZDW', where Z is the eigenvectors of AA' and W is the eigenvectors of A'A

/* and the D is a diagonal but not square matrix.

matrix.

compute A={3,1,1;-1,3,1}./* see page 32 Lattin et al. 

print a.

call svd(A,z,d,w).

print z.

print d.

print w.

compute a_p=z*d*t(w).

print a_p.

end matrix.

Run MATRIX procedure:

A

  3  1  1

 -1  3  1

Z

   .7071067812  -.7071067812

   .7071067812   .7071067812

D

   3.464101615    .000000000    .000000000

    .000000000   3.162277660    .000000000

W

   .4082482905  -.8944271910   .1825741858

   .8164965809   .4472135955   .3651483717

   .4082482905   .0000000000  -.9128709292

A_P

   3.000000000   1.000000000   1.000000000

  -1.000000000   3.000000000   1.000000000

------ END MATRIX -----

